



1) AUDIO FREQ. NON-INVERTING AMPLIFIER



FOR FREQ > 33HZ, $V_{OUT} \cong 1.5 V_{IN}$

2) ALL PASS FILTER (CONSTANT GAIN, PHASE LEAD)



a) $V_{OUT} = V_{IN}$ FOR ALL FREQ,


b) ANGLE OF $V_{OUT} =$ ANGLE OF $V_{IN} + \theta^\circ$

WHERE $180^\circ \leq \theta^\circ \leq 0^\circ$


BETWEEN $\frac{1}{10} f_1 - 10 f_1$

AND 90° AT f_1


3) SUMMING CIRCUIT



4) ELECTRONIC VARIABLE RESISTOR




5) TRIANGLE-WAVE OSCILLATOR



FOR $\text{SPEED} = 500\text{K}\Omega$, $f = 0.14\text{Hz}$; $\text{SPEED} = 1\Omega$, $f = 8\text{Hz}$


6) SPLIT SUPPLY




- OP-AMPS NEED \pm POWER SUPPLIES TO WORK BUT WE CAN BIAS THE OP-AMPS TO CREATE THE SAME EFFECT

7) FINAL FILTER


- WE NEED TO BLOCK THIS DC BIASING FROM OUR POWER AMPLIFIERS BECAUSE IT WILL JUST PULL OR PUSH THE SPEAKER CONE.



$$V_{OUT} \cong V_{IN} \text{ FOR FREQ. } > 21.2 \text{ KHz}$$



- PSPICE NODES



- INPUT FILE :

```

MXR Phase 45
VS 1 0 SIN (0 1 3K)
VCC 5 0 9
*
R23 2 3 10K
R57 5 7 10K
R47 4 7 20K
R37 3 7 470K
R46 4 6 10K
R87 8 7 10K
R810 8 10 10K
R915 9 15 470K
R713 7 13 10K
R611 6 11 10K
R1112 11 12 10K
R1218 12 18 10K
R1317 13 17 10K
R1415 14 15 470K
R1516 15 16 1MEG
R1819 18 19 10K
R1920 19 20 10K
R620 6 20 10K
R210 21 0 150K
R1522 15 22 3.9MEG
R2223 22 23 150K
R524 5 24 150K
R240 24 0 150K
R2425 24 25 150K
R2526 25 26 7.5K
*
* POTS
R716 7 16 125K
R160 16 0 125K
*
R2226 22 26 50K
*
C50 5 0 10U
C70 7 0 10U
C220 22 0 10U
C12 1 2 .01U
C89 8 9 .01U
C1314 13 14 .01U
C2325 23 25 .01U
C715 7 15 .05U
C610 6 10 .05U
C1217 12 17 .05U
C2021 20 21 .05U
*
J1 10 9 7 J2N5952
J2 17 14 7 J2N5952
*
D1 0 7 D1N5230
*
X1 3 4 5 0 6 TL074
X2 10 11 5 0 12 TL074
X3 17 18 5 0 19 TL074
X4 24 23 5 0 25 TL074
*
* TL074 is 1/2-TL072
*
.LIB PHASE45.LIB
.TRAN 2U 2000M 0 2U
.PROBE
.END

```

- MODEL LIBRARY USED



```

* PHASE45.LIB - library file for models
*-----
.model D1N5230 D(Is=880.5E-18 Rs=.25 Ikf=0 N=1
+ Xti=3 Eg=1.11 Cjo=175p M=.5516 Vj=.75 Fc=.5
+ Isr=1.859n Nr=2 Bv=4.7 Ibv=20.245m Nbv=1.6989
+ Ibv1=1.9556m Nbv1=14.976 Tbv1=-21.28u)
* Motorola pid=1N5230 case=DO-35
* 89-9-18 gjg
* Vz = 4.7 @ 20mA, Zz = 300 @ 1mA,
* Zz = 12.5 @ 5mA, Zz =2.6 @ 20m
*-----
.model J2N5952 NJF(Beta=1.314m Betatce=-.5 Rd=1 Rs=1
+ Lambda=4m Vto=-2.021 Vtotc=-2.5m Is=33.57f
+ Isr=322.4f N=1 Nr=2 Xti=3 Alpha=311.7u
+ Vk=243.6 Cgd=1.6p M=.3622 Pb=1 Fc=.5 Cgs=2.414p
+ Kf=10.14E-18 Af=1)
* National pid=50 case=TO92
* 88-08-02 rnm BVmin=30
*-----
* connections: non-inverting input
* | inverting input
* | | positive power supply
* | | | negative power supply
* | | | | output
* | | | | |
.subckt TL074 1 2 3 4 5
*
c1 11 12 2.412E-12
c2 6 7 18.00E-12
css 10 99 5.400E-12
dc 5 53 dx
de 54 5 dx
dlp 90 91 dx
dln 92 90 dx
dp 4 3 dx
egnd 99 0 poly(2) (3,0) (4,0) 0 .5 .5
fb 7 99 poly(5) vb vc ve vlp vln
+ 0 3.467E6 -3E6 3E6 3E6 -3E6
ga 6 0 11 12 339.3E-6
gcm 0 6 10 99 17.01E-9
iss 10 4 dc 234.0E-6
hlim 90 0 vlim 1K
j1 11 2 10 jx
j2 12 1 10 jx
r2 6 9 100.0E3
rd1 3 11 2.947E3
rd2 3 12 2.947E3
ro1 8 5 50
ro2 7 99 170
rp 3 4 20.00E3
rss 10 99 854.7E3
vb 9 0 dc 0
vc 3 53 dc 1.500
ve 54 4 dc 1.500
vlim 7 8 dc 0
vlp 91 0 dc 50
vln 0 92 dc 50
.model dx D(Is=800.0E-18 Rs=1)
.model jx NJF(Is=2.500E-12 Beta=984.2E-6 Vto=-1)
.ends
*

```


MXR Phase 45,...

Temperature: 27.0,...



MXR Phase 45

Temperature: 27.0





- THE FREQ. WHERE THE ANGLE OF THE ALL-PASS FILTER IS 90° IS

$$f_1 = \frac{1}{2\pi \cdot R_{DS} \cdot 0.05\mu} \approx 3\text{KHZ FOR } R_{DS} = 1.08\text{K}\Omega$$

- THUS WE HAVE A TOTAL PHASE SHIFT OF 180° FROM THE TWO ALL-PASS FILTERS



MXR Phase 45,...

Temperature: 27.0,...



MXR Phase 45,...

Temperature: 27.0,...





- AT 0.741 SEC, $R_{DS} = 562$, $f_i \approx 6\text{KHZ}$, $\theta \approx 120^\circ$, $2\theta = 240^\circ$

- RE-RUN WITH $V_{IN} = 1V$ SINEWAVE @ 6KHZ

MXR Phase 45,...

Temperature: 27.0,...




- NOW THE 6KHZ TONE IS CANCELED

- RE-RUN WITH $V_{IN} = 1V$ SINEWAVE @ 9KHZ

MXR Phase 45

Temperature: 27.0



- NOW THERE IS NO CANCELLATION. WHY?

- $500\Omega \leq R_{DS} \leq 1.6K\Omega$ ← SEE P7

USING $f_1 = \frac{1}{2\pi R_{DS}(0.05\mu)}$

THEN THE FREQ. WHERE WE GET 180° OF PHASE-SHIFT, AND THUS CANCELLATION, IS

$$2KHZ \leq f_1 \leq 6.4KHZ$$